<commit_before>if not set -q shellar
  set shellar $HOME/.shellar
end

. $shellar/shellar/lib.fish

# load library files
function shellar_load_configs
  for config in $shellar_lib_dir/*.fish
    . $config
  end
end

# load plugins
function shellar_load_plugins
  for plugin in $shellar_plugins
    set -l p $shellar_plugins_dir/$plugin/init.fish
    set -l c $shellar_custom_plugin_dir/$plugin/init.fish
    if test -f $c
      . $c
    else if test -f $p
      . $p
    end
  end
end

# load theme
function shellar_load_themes
  if set -q shellar_theme
    . $shellar_themes_dir/$shellar_theme/$shellar_theme.theme.fish
  else
    . $shellar_themes_dir/default/default.theme.fish
  end
end

# add functions dir to function path
function shellar_load_functions
  for plugin in $shellar_plugins
    set -l f $shellar_plugins_dir/$plugin/functions
    if test -d $f
      set fish_function_path $f $fish_function_path
    end
  end
end

shellar_load_configs
shellar_load_plugins
shellar_load_themes
shellar_load_functions<commit_msg> fix loading of functions directory<commit_after>if not set -q shellar
  set shellar $HOME/.shellar
end

. $shellar/shellar/lib.fish

# load library files
function shellar_load_configs
  for config in $shellar_lib_dir/*.fish
    . $config
  end
end

# load plugins
function shellar_load_plugins
  for plugin in $shellar_plugins
    set -l p $shellar_plugins_dir/$plugin/init.fish
    set -l c $shellar_custom_plugin_dir/$plugin/init.fish
    if test -f $c
      . $c
    else if test -f $p
      . $p
    end
  end
end

# load theme
function shellar_load_themes
  if set -q shellar_theme
    . $shellar_themes_dir/$shellar_theme/$shellar_theme.theme.fish
  else
    . $shellar_themes_dir/default/default.theme.fish
  end
end

# add functions dir to function path
function shellar_load_functions
  for plugin in $shellar_plugins
    set -l f $shellar_plugins_dir/$plugin/functions
    set -l c $shellar_custom_plugin_dir/$plugin/functions
    if test -d $c; and not contains $c $fish_function_path
      set -x fish_function_path $c $fish_function_path
    else if test -d $f; and not contains $f $fish_function_path
      set -x fish_function_path $f $fish_function_path
    end
  end
end

shellar_load_configs
shellar_load_plugins
shellar_load_themes
shellar_load_functions
{'fraction_non_alphanumeric': 0.06637744034707159, 'fraction_numerical': 0.0, 'mean_word_length': 3.512720156555773, 'pattern_counts': {'":': 0, '<': 3, '<?xml version=': 0, '>': 3, 'https://': 0, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '3669398', 'n_tokens_mistral': 799, 'n_tokens_neox': 790, 'n_words': 221}